

Design Recipes

April 25, 2021

Contents

1	Design Recipes	4
2	How To Design Functions (HtDF)	6
2.1	Signature, purpose and stub.	7
2.2	Define examples, wrap each one in check-expect.	7
2.3	Template and inventory	8
2.4	Code the function body	9
2.5	Test and debug until correct	10
3	How To Design Data (HTDD)	11
3.1	What is the Inherent Structure of the Information?	13
3.2	Simple Atomic Data	14
3.3	Intervals	15
3.4	Enumerations	16
3.5	Itemizations	18
3.6	Compound data (structures)	20
3.7	References to other data definitions	21
3.8	Self-referential or mutually referential	23
4	How To Design Worlds (HtDW)	25
4.1	Phase 1: Domain Analysis	26
4.2	Phase 2: Building the actual program	27
4.3	Template for a World Program	27
4.4	Key and Mouse Handlers	30
5	Data Driven Templates	31
5.1	Producing the Template for an Example - One Of Type	37
5.2	Templates for Mutually Referential Types	38
5.3	Testing	39
5.4	Additional Design Rules for Helpers	39
6	Functions On 2 One-Of Data	40

7	Control Driven Templates	44
7.1	Function Composition	44
7.2	Backtracking Search	45
7.3	Generative Recursion	47
8	Accumulators	48
8.1	The basic recipe and context preserving accumulators.	48
8.2	Tail Recursion	52
9	Template Blending	53
10	Abstraction	58
10.1	Abstraction From Examples	58
10.2	Abstraction From Type Comments	59
10.3	Using Abstract Functions	61

List of Tables

1.1	Types of Design Recipes	5
3.1	Types of Data Definition	13
4.1	Big-Bang Options	26
5.1	Data Driven Templates	36

Chapter 1

Design Recipes

In this course, we teach an approach to program design based on design recipes. Each recipe is applicable to certain problems, and systematizes the process of designing solutions to those problems.

There are three core recipes that are used most frequently. The **templating recipes** are used as part of the design of every data definition and function. **Abstraction recipes** are used to reduce redundancy in code.

Core Recipes	Templating		Abstraction
	Data Driven	Control Driven	
<i>How To Design Functions (HtDF)</i> Design any function.	<i>Data Driven Templates</i> Produce template for a data definition based on the form of the type comment.	<i>Function Composition</i>	<i>Abstraction From Examples</i> Produce an abstract function given two similar functions.
<i>How To Design Data (HTDD)</i> Produce data definitions based on structure of the information to be represented.	<i>Functions On 2 One-Of Data</i> functions where 2 arguments have a one-of in their type comments.	<i>Backtracking Search</i>	<i>Abstraction From Type Comments</i> Produce a fold function given type comments.
<i>How To Design Worlds (HtDW)</i> Produce interactive programs that use big-bang.		<i>Generative Recursion</i>	
		<i>Accumulators</i>	<i>Using Abstract Functions</i>
	<i>Template Blending</i>		

Table 1.1: Types of Design Recipes

Chapter 2

How To Design Functions (HtDF)

The How to Design Functions (HtDF) recipe is a **design method** that enables systematic design of functions. We will use this recipe throughout the term, although we will enhance it as we go to solve more complex problems.

The HtDF recipe consists of the following steps:

1. Signature, purpose and stub.
2. Define examples, wrap each in check-expect.
3. Template and inventory.
4. Code the function body.
5. Test and debug until correct

NOTE:

- Each of these steps build on the ones that precede it. The signature helps write the purpose, the stub, and the check-expects; it also helps code the body. The purpose helps write the check-expects and code the body. The stub helps to write the check-expects. The check-expects help to code the body as well as to test the complete design.
- It is sometimes helpful to do the steps in a different order. Sometimes it is easier to write examples first, then do signature and purpose. Often at some point during the design you may discover an issue or boundary condition you did not anticipate, at that point go back and update the purpose and examples accordingly. But you should never write the function definition first and then go back and do the other recipe elements – for some of you that will work for simple functions,

but you will not be able to do that for the complex functions later in the course!

- Throughout the HtDF process be sure to "run early and run often". Run your program whenever it is well- formed. The more often you press run the sooner you can find mistakes. Finding mistakes one at a time is much easier than waiting until later when the mistakes can compound and be more confusing. Run, run, run!

2.1 Signature, purpose and stub.

Write the function signature, a one-line purpose statement and a function stub.

A signature has the type of each argument, separated by spaces, followed by `->`, followed by the type of result. So a function that consumes an image and produces a number would have the signature `Image -> Number`.

Note that the stub is a syntactically complete function definition that produces a value of the right type. If the type is `Number` it is common to use `0`, if the type is `String` it is common to use `"a"` and so on. The value will not, in general, match the purpose statement. In the example below the stub produces `0`, which is a `Number`, but only matches the purpose when `double` happens to be called with `0`.

```
1  ;; Number -> Number
2  ;; produces n times 2
3  (define (double n) 0) ; this is the stub
```

The purpose of the stub is to serve as a kind of scaffolding to make it possible to run the examples even before the function design is complete. With the stub in place check-expects that call the function can run. Most of them will fail of course, but the fact that they can run at all allows you to ensure that they are at least well-formed: parentheses are balanced, function calls have the proper number of arguments, function and constant names are correct and so on. This is very important, the sooner you find a mistake – even a simple one – the easier it is to fix.

2.2 Define examples, wrap each one in check-expect.

Write at least one example of a call to the function and the expected result the call should produce.

You will often need more examples, to help you better understand the function or to properly test the function. If once your function works and you run the program some of the code is highlighted in black it means you definitely do not have enough examples. If you are unsure how to start

writing examples use the combination of the function signature and the data definition(s) to help you generate examples. Often the example data from the data definition is useful, but it does not necessarily cover all the important cases for a particular function.

The first role of an example is to help you understand what the function is supposed to do. If there are boundary conditions be sure to include an example of them. If there are different behaviours the function should have, include an example of each. Since they are examples first, you could write them in this form:

```
1  ;; (double 0) should produce 0
2  ;; (double 1) should produce 2
3  ;; (double 2) should produce 4
```

When you write examples it is sometimes helpful to write not just the expected result, but also how it is computed. For example, you might write the following instead of the above:

```
1  ;; (double 0) should produce (* 0 2)
2  ;; (double 1) should produce (* 1 2)
3  ;; (double 2) should produce (* 2 2)
```

While the above form satisfies our need for examples, DrRacket gives us a better way to write them, by enclosing them in `check-expect`. This will allow DrRacket to check them automatically when the function is complete. (In technical terms it will turn the examples into unit tests.)

```
1  ;; Number -> Number
2  ;; produces n times 2
3  (check-expect (double 0) (* 0 2))
4  (check-expect (double 1) (* 1 2))
5  (check-expect (double 3) (* 3 2))
6
7  (define (double n) 0); this is the stub
```

The existence of the stub will allow you to run the tests. Most (or even all) of the tests will fail since the stub is returning the same value every time. But you will at least be able to check that the parentheses are balanced, that you have not misspelled function names etc.

2.3 Template and inventory

Before coding the function body it is helpful to have a clear sense of what the function has to work with – what is the contents of your bag of parts for coding this function? The template provides this.

Once the How to Design Data Definitions (HtDD) recipe is introduced, templates are produced by following the rules on the Data Driven Templates section. You should copy the template from the data definition to the function design, rename the template, and write a comment that says where the template was copied from. Note that the template is copied from the data definition for the consumed type, not the produced type.

For primitive data like numbers, strings and images the body of the template is simply `(... x)` where `x` is the name of the parameter to the function.

Once the template is done the stub should be commented out.

```

1  ;; Number -> Number
2  ;; produces n times 2
3  (check-expect (double 0) (* 0 2))
4  (check-expect (double 1) (* 1 2))
5  (check-expect (double 3) (* 3 2))
6
7  ;(define (double n) 0) ; this is the stub
8
9  (define (double n)      ; this is the template
10     (... n))

```

It is also often useful to add constant values which are extremely likely to be useful to the template body at this point. For example, the template for a function that renders the state of a world program might have an MTS constant added to its body. This causes the template to include an inventory of useful constants.

2.4 Code the function body

Now complete the function body by filling in the template.

Note that:

- the signature tells you the type of the parameter(s) and the type of the data the function body must produce
- the purpose describes what the function body must produce in English
- the examples provide several concrete examples of what the function body must produce
- the template tells you the raw material you have to work with

You should use all of the above to help you code the function body. In some cases further rewriting of examples might make it more clear how you computed certain values, and that may make it easier to code the function.

```

1  ;; Number -> Number
2  ;; produces n times 2
3  (check-expect (double 0) (* 0 2))
4  (check-expect (double 1) (* 1 2))
5  (check-expect (double 3) (* 3 2))
6
7  ;(define (double n) 0) ; this is the stub
8
9  ;(define (double n)    ; this is the template
10 ;    (... n))
11
12 (define (double n)
13     (* n 2))

```

2.5 Test and debug until correct

Run the program and make sure all the tests pass, if not debug until they do. Many of the problems you might have had will already have been fixed because of following the "run early, run often" approach. But if not, debug until everything works.

Chapter 3

How To Design Data (HTDD)

Data definitions are a driving element in the design recipes.

A data definition establishes the represent/interpret relationship between information and data:

- **Information** in the program's domain is represented by data in the program.
- **Data** in the program can be interpreted as information in the program's domain.

A data definition must describe how to form (or make) data that satisfies the data definition and also how to tell whether a data value satisfies the data definition. It must also describe how to represent information in the program's domain as data and interpret a data value as information.

So, for example, one **data definition** might say that numbers **are used to represent** the Speed of a ball. Another data definition might say that numbers are used to represent the Height of an airplane. So given a number like 6, **we need a data definition to tell us how to interpret it:** is it a Speed, or a Height or something else entirely. Without a data definition, the 6 could mean anything.

The first step of the recipe is to identify the inherent structure of the information.

Once that is done, a data definition consists of four or five elements:

1. A possible *structure definition* (not until compound data)
2. A *type comment* that defines a new type name and describes how to form data of that type.
3. An *interpretation* that describes the correspondence between information and data.
4. One or more *examples* of the data.
5. A *template* for a 1 argument function operating on data of this type.

In the first weeks of the course we also ask you to include a list of the *template rules* used to form the template.

3.1 What is the Inherent Structure of the Information?

One of the most important points in the course is that:

- the *structure of the information* in the program's domain determines the kind of data definition used,
- which in turn determines the *structure of the templates* and helps determine the function examples (check-expects),
- and therefore the *structure of much of the final program design*.

The remainder of this chapter lists in detail different kinds of data definition that are used to represent information with different structures. It also shows in detail how to design a data definition of each kind. This summary table provides a quick reference to which kind of data definition to use for different information structures.

When the form of the information to be represented...	Use a data definition of this kind
is atomic	Simple Atomic Data
is numbers within a certain range	Intervals
consists of a fixed number of distinct items	Enumerations
is comprised of 2 or more subclasses, at least one of which is not a distinct item	Itemizations
consists of two or more items that naturally belong together	Compound data (structures)
is naturally composed of different parts	References to other data definitions
is of arbitrary (unknown) size	Self-referential or mutually referential

Table 3.1: Types of Data Definition

3.2 Simple Atomic Data

Use simple atomic data when the information to be represented is itself atomic in form, such as the elapsed time since the start of the animation, the x coordinate of a car or the name of a cat.

```
1  ;; Time is Natural
2  ;; interp. number of clock ticks
3  ;; since start of game
4  (define START-TIME 0)
5  (define OLD-TIME 1000)
6
7  #;
8  (define (fn-for-time t)
9          (... t))
10
11  ;; Template rules used:
12  ;; - atomic non-distinct: Natural
```

Forming the Template

As noted below the template, it is formed according to the Data Driven Templates recipe using the right hand column of the atomic non-distinct rule.

Guidance on Data Examples and Function Example/Tests

One or two data examples are usually sufficient for simple atomic data.

When creating example/tests for a specific function operating on simple atomic data at least one test case will be required. Additional tests are required if there are multiple cases involved. If the function produces Boolean there needs to be at least a true and false test case. Also be on the lookout for cases where a number of some form is an interval in disguise, for example given a type comment like Countdown is Natural, in some functions 0 is likely to be a special case.

3.3 Intervals

Use an interval when the information to be represented is numbers within a certain range. `Integer[0, 10]` is all the integers from 0 to 10 inclusive; `Number[0, 10)` is all the numbers from 0 inclusive to 10 exclusive. The notation is that `[` and `]` mean that the end of the interval includes the end point; `(` and `)` mean that the end of the interval does not include the end point.)

Intervals often appear in itemizations, but can also appear alone, as in:

```
1  ;; Countdown is Integer[0, 10]
2  ;; interp. the number of seconds
3  ;; remaining to liftoff
4  (define C1 10) ; start
5  (define C2 5)  ; middle
6  (define C3 0)  ; end
7
8  #;
9  (define (fn-for-countdown cd)
10         (... cd))
11
12  ;; Template rules used:
13  ;; - atomic non-distinct: Integer[0, 10]
```

Forming the Template

As noted below the template, it is formed according to the Data Driven Templates recipe using the right hand column of the atomic non-distinct rule.

Guidance on Data Examples and Function Example/Tests

For data examples provide sufficient examples to illustrate how the type represents information. The three data examples above are probably more than is needed in that case.

When writing tests for functions operating on intervals **be sure to test closed boundaries as well as midpoints**. As always, be sure to include enough tests to check all other points of variance in behaviour across the interval.

3.4 Enumerations

Use an enumeration *when the information to be represented consists of a fixed number of distinct items*, such as colors, letter grades etc. The data used for an enumeration could in principle be anything - strings, integers, images even. But we always use strings. In the case of enumerations it is sometimes redundant to provide an interpretation and nearly always redundant to provide examples. The example below includes the interpretation but not the examples.

```
1  ;; LightState is one of:
2  ;; - "red"
3  ;; - "yellow"
4  ;; - "green"
5  ;; interp. the color of a traffic light
6
7  ;; <examples are redundant for enumerations>
8
9  #;
10 (define (fn-for-light-state ls)
11     (cond [(string=? "red" ls) (...)]
12           [(string=? "yellow" ls) (...)]
13           [(string=? "green" ls) (...)]))
14
15 ;; Template rules used:
16 ;; - one of: 3 cases
17 ;; - atomic distinct: "red"
18 ;; - atomic distinct: "yellow"
19 ;; - atomic distinct: "green"
```

Forming the Template

As noted below the template, it is formed according to the Data Driven Templates recipe as follows:

First, LightState is an enumeration with 3 cases, so the one of rule says to use a cond with 3 cases:

```
1  (define (fn-for-tlcolor ls)
2      (cond [Q1 A1]
3            [Q2 A2]
4            [Q3 A3]))
```

In the first clause, "red" is a distinct atomic value, so the cond question column of the atomic distinct rule says Q1 should be (string=? ls "red"). The cond answer column says A1 should be (...). So we have:

```

1 (define (fn-for-light-state ls)
2   (cond [(string=? "red" ls) (...)]
3         [Q2 A2]
4         [Q3 A3]))

```

Then "yellow" and "green" are also distinct atomic values, so the final template is:

```

1 (define (fn-for-light-state ls)
2   (cond [(string=? "red" ls) (...)]
3         [(string=? "yellow" ls) (...)]
4         [(string=? "green" ls) (...)]))

```

Guidance on Data Examples and Function Example/Tests

Data examples are redundant for enumerations.

Functions operating on enumerations should have (at least) as many tests as there are cases in the enumeration.

Large Enumerations

Some enumerations contain a large number of elements. A canonical example is KeyEvent, which is provided as part of big-bang. KeyEvent includes all the letters of the alphabet as well as other keys you can press on the keyboard. It is not necessary to write out all the cases for such a data definition. Instead write one or two, as well as a comment saying what the others are, where they are defined etc.

Defer writing templates for such large enumerations until a template is needed for a specific function. At that point include the specific cases that function cares about. Be sure to include an else clause in the template to handle the other cases. As an example, some functions operating on KeyEvent may only care about the space key and just ignore all other keys, the following would be an appropriate template for such functions.

```

1 #;
2 (define (fn-for-key-event kevt)
3   (cond [(key=? " " kevt) (...)]
4         [else
5          (...)]))
6
7 ;; Template formed using the large
8 ;; enumeration special case

```

The same is true of writing tests for functions operating on large enumerations. All the specially handled cases must be tested, in addition one more test is required to check the else clause.

3.5 Itemizations

An itemization describes *data comprised of 2 or more subclasses, at least one of which is not a distinct item*. (C.f. enumerations, where the subclasses are *all* distinct items.) In an itemization the template is similar to that for enumerations: a cond with one clause per subclass. In cases where the subclass of data has its own data definition the answer part of the cond clause includes a call to a helper template, in other cases it just includes the parameter.

```
1  ;; Bird is one of:
2  ;; - false
3  ;; - Number
4  ;; interp. false means no bird,
5  ;; number is x position of bird
6
7  (define B1 false)
8  (define B2 3)
9
10 #;
11 (define (fn-for-bird b)
12       (cond [(false? b) (...)]
13             [(number? b) (... b)]))
14
15 ;; Template rules used:
16 ;; - one of: 2 cases
17 ;; - atomic distinct: false
18 ;; - atomic non-distinct: Number
```

Forming the Template

As noted below the template, it is formed according to the Data Driven Templates recipe using the *one-of rule*, the *atomic distinct rule* and the *atomic non-distinct rule* in order.

Guidance on Data Examples and Function Example/Tests

As always, itemizations should have enough data examples to clearly illustrate how the type represents information.

Functions operating on itemizations should have *at least as many tests as there are cases in the itemizations*. If there are intervals in the itemization, then there should be tests at all points of variance in the interval. In the case of adjoining intervals it is critical to test the boundaries.

Itemization of Intervals

A common case is for the itemization to be comprised of 2 or more intervals. In this case functions operating on the data definition will usually need to be tested at all the boundaries of closed intervals and points between the boundaries.

```
1  ;;; Reading is one of:
2  ;; - Number[> 30]
3  ;; - Number(5, 30]
4  ;; - Number[0, 5]
5  ;; interp. distance in cm from bumper to obstacle
6  ;; Number[> 30]          is considered "safe"
7  ;; Number(5, 30]        is considered "warning"
8  ;; Number[0, 5]         is considered "dangerous"
9
10 (define R1 40)
11 (define R2 .9)
12
13 #;
14 (define (fn-for-reading r)
15   (cond [(< 30 r) (... r)]
16         [(and (< 5 r) (<= r 30)) (... r)]
17         [(<= 0 r 5) (... r)]))
18
19 ;; Template rules used:
20 ;; one-of: 3 cases
21 ;; atomic non-distinct: Number[>30]
22 ;; atomic non-distinct: Number(5, 30]
23 ;; atomic non-distinct: Number[0, 5]
```

As noted below the template, it is formed according to the Data Driven Templates recipe using the *one-of rule*, followed by 3 uses of the *atomic non-distinct rule*.

3.6 Compound data (structures)

Use structures when two or more values naturally belong together. The `define-struct` goes at the beginning of the data definition, before the types comment.

```
1 (define-struct ball (x y))
2 ;; Ball is (make-ball Number Number)
3 ;; interp. a ball at position x, y
4
5 (define BALL-1 (make-ball 6 10))
6
7 #;
8 (define (fn-for-ball b)
9   (... (ball-x b)      ;Number
10        (ball-y b)))   ;Number
11
12 ;; Template rules used:
13 ;; - compound: 2 fields
```

The template above is formed according to the Data Driven Templates recipe using the compound rule. Then for each of the selectors, the result type of the selector (Number in the case of `ball-x` and `ball-y`) is used to decide whether the selector call itself should be wrapped in another expression. In this case, where the result types are primitive, no additional wrapping occurs. C.f. cases below when the reference rule applies.

Guidance on Data Examples and Function Example/Tests

For compound data definitions it is often useful to have numerous examples, for example to illustrate special cases. For a snake in a snake game you might have an example where the snake is very short, very long, hitting the edge of a box, touching food etc. These data examples can also be useful for writing function tests because they save space in each check-expect.

3.7 References to other data definitions

Some data definitions contain references to other data definitions you have defined (non-primitive data definitions). One common case is for a compound data definition to comprise other named data definitions. (Or, once lists are introduced, for a list to contain elements that are described by another data definition. In these cases the template of the first data definition should contain calls to the second data definition's template function wherever the second data appears. For example:

```
1  ;---assume Ball is as defined above---
2
3  (define-struct game (ball score))
4  ;; Game is (make-game Ball Number)
5  ;; interp. the current ball and score of the game
6
7  (define GAME-1 (make-game (make-ball 1 5) 2))
8
9  #;
10 (define (fn-for-game g)
11         (... (fn-for-ball (game-ball g))
12              (game-score g))) ;Number
13
14 ;; Template rules used:
15 ;; - compound: 2 fields
16 ;; - reference: ball field is Ball
```

In this case the template is formed according to the Data Driven Templates recipe by first using the compound rule. Then, since the result type of (game-ball g) is Ball, the reference rule is used to wrap the selector so that it becomes (fn-for-ball (game-ball g)). The call to game-score is not wrapped because it produces a primitive type.

Guidance on Data Examples and Function Example/Tests

For data definitions involving references to non-primitive types the data examples can sometimes become quite long. In these cases it can be helpful to define well-named constants for data examples for the referred to type and then use those constants in the referring from type. For example:

```
1  ;...in the data definition for Drop...
2  (define DTOP (make-drop 10 0)) ;top of
   ↪ screen
3  (define DMID (make-drop 20 (/ HEIGHT 2))) ;middle of
   ↪ screen
```

```

4      (define DBOT (make-drop 30 HEIGHT))           ;at bottom
      ↪ edge
5      (define DOUT (make-drop 40 (+ HEIGHT 1)))     ;past
      ↪ bottom edge
6
7      ;...in the data definition for ListOfDrop...
8      (define LOD1 empty)
9      (define LOD-ALL-ON          (cons DTOP (cons DMID
      ↪ empty)))
10     (define LOD-ONE-ABOUT-TO-LEAVE (cons DTOP
11                                         (cons DMID (cons DBOT
      ↪ empty))))
12     (define LOD-ONE-OUT-ALREADY  (cons DTOP (cons DMID
13                                         (cons DBOT (cons DOUT
14                                                         empty))))))

```

In the case of references to non-primitive types the function operating on the referring type (i.e. ListOfDrop) will end up with a call to a helper that operates on the referred to type (i.e. Drop). Tests on the helper function should fully test that function, tests on the calling function may assume the helper function works properly.

3.8 Self-referential or mutually referential

When the *information in the program's domain is of arbitrary size*, a well-formed self-referential (or mutually referential) data definition is needed.

In order to be well-formed, a self-referential data definition must:

- have at least one case without self reference (the base case(s))
- have at least one case with self reference

The *template* contains a base case corresponding to the non-self-referential clause(s) as well as one or more natural recursions corresponding to the self-referential clauses.

```
1  ;; ListOfString is one of:
2  ;; - empty
3  ;; - (cons String ListOfString)
4  ;; interp. a list of strings
5
6  (define LOS-1 empty)
7  (define LOS-2 (cons "a" empty))
8  (define LOS-3 (cons "b" (cons "c" empty)))
9
10 #;
11 (define (fn-for-los los)
12     (cond [(empty? los) (...)] ;BASE CASE
13           [else (... (first los) ;String
14                       (fn-for-los (rest los)))] ;NATURAL
15             ↪ RECURSION
16           )
17 ;;
18 ;;
19 ;; COMBINATION
20 ;; Template rules used:
21 ;; - one of: 2 cases
22 ;; - atomic distinct: empty
23 ;; - compound: (cons String ListOfString)
24 ;; - self-reference: (rest los) is ListOfString
```

In some cases a types comment can have both self-reference and reference to another type.

```
1  (define-struct dot (x y))
2  ;; Dot is (make-dot Integer Integer)
3  ;; interp. A dot on the screen, w/ x and y coordinates.
```



```

4
5 (define D1 (make-dot 10 30))
6
7 #;
8 (define (fn-for-dot d)
9       (... (dot-x d)           ;Integer
10            (dot-y d)))        ;Integer
11
12 ;; Template rules used:
13 ;; - compound: 2 fields
14
15
16 ;; ListOfDot is one of:
17 ;; - empty
18 ;; - (cons Dot ListOfDot)
19 ;; interp. a list of Dot
20
21 (define LOD1 empty)
22 (define LOD2 (cons (make-dot 10 20)
23                   (cons (make-dot 3 6) empty)))
24
25 #;
26 (define (fn-for-lod lod)
27       (cond [(empty? lod) (...)]
28             [else
29              (... (fn-for-dot (first lod))
30                   (fn-for-lod (rest lod)))]))
31
32 ;; Template rules used:
33 ;; - one of: 2 cases
34 ;; - atomic distinct: empty
35 ;; - compound: (cons Dot ListOfDot)
36 ;; - reference: (first lod) is Dot
37 ;; - self-reference: (rest lod) is ListOfDot

```

Guidance on Data Examples and Function Example/Tests

When writing data and function examples for self-referential data definitions always put the base case first. Its usually trivial for data examples, but many function tests don't work properly if the base case isn't working properly, so testing that first can help avoid being confused by a failure in a non base case test. Also be sure to have a test for a list (or other structure) that is at least 2 long.

Chapter 4

How To Design Worlds (HtDW)

The How to Design Worlds process provides guidance for **designing interactive world programs using big-bang**. While some elements of the process are tailored to big-bang, the process can also be adapted to the design of other interactive programs. The wish-list technique can be used in any multi-function program.

World program design is divided into two phases, each of which has sub-parts:

1. Domain analysis (use a piece of paper!)
 - (a) Sketch program scenarios
 - (b) Identify constant information
 - (c) Identify changing information
 - (d) Identify big-bang options
2. Build the actual program
 - (a) Constants (based on item 1b above)
 - (b) Data definitions using [How To Design Data \(HTDD\)](#) (based on item 1c above)
 - (c) Functions using [How To Design Functions \(HtDF\)](#)
 - i. main first (based on item 1c, item 1d and item 2b above)
 - ii. wish list entries for big-bang handlers
 - (d) Work through wish list until done

4.1 Phase 1: Domain Analysis

Do a domain analysis by hand-drawing three or more pictures of what the world program will look like at different stages when it is running.

Use this picture to identify constant information such as the height and width of screen, color of the background, the background image itself, the length of a firework's fuse, the image for a moving cat and so on.

Also identify changing information such as the position of a firework, the color of a light, the number in countdown etc.

Identify which big-bang options the program needs.

If your program needs to:	Then it needs this option:
change as time goes by (nearly all do)	on-tick
display something (nearly all do)	to-draw
change in response to key presses	on-key
change in response to mouse activity	on-mouse
stop automatically	stop-when

Table 4.1: Big-Bang Options

(There are several more options to big-bang. Look in the DrRacket help desk under big-bang for a complete list.)

4.2 Phase 2: Building the actual program

Structure the actual program in four parts:

- Requires followed by one line summary of program's behavior
- Constants
- Data definitions
- Functions

The program should begin with whatever require declarations are required. For a program using big-bang this is usually a require for 2htdp/universe to get big-bang itself and a require for 2htdp/image to get useful image primitives. This is followed by a short summary of the program's behavior (ideally 1 line).

The next section of the file should define constants. These will typically come directly from the domain analysis.

This is followed by data definitions. The data definitions describe how the world state - the changing information identified during the analysis - will be represented as data in the program. Simple world programs may have just a single data definition. More complex world programs have a number of data definitions.

The functions section should begin with the main function which uses big-bang with the appropriate options identified during the analysis. After that put the more important functions first followed by the less important helpers. Keep groups of closely related functions together.

4.3 Template for a World Program

A useful template for a world program, including a template for the main function and wish list entries for tick-handler and to-draw handler is as follows. To use this template replace WS with the appropriate type for your changing world state. You may want to give the handler functions more descriptive names and you should definitely give them all a more descriptive purpose.

```
1 (require 2htdp/image)
2 (require 2htdp/universe)
3
4 ;; My world program (make this more specific)
5
6 ;; =====
7 ;; Constants:
8
```

```

9  ;; =====
10 ;; Data definitions:
11 ;; WS is ... (give WS a better name)
12
13 ;; =====
14 ;; Functions:
15 ;; WS -> WS
16 ;; start the world with ...
17 ;;
18 (define (main ws)
19   (big-bang ws          ;WS
20     (on-tick tock)      ;WS -> WS
21     (to-draw render)   ;WS -> Image
22     (stop-when ...)    ;WS -> Boolean
23     (on-mouse ...)     ;WS Integer Integer MouseEvent ->
      ↪ WS
24     (on-key ...)))    ;WS KeyEvent -> WS
25
26 ;; WS -> WS
27 ;; produce the next ...
28 ;; !!!
29 (define (tock ws) ...)
30
31 ;; WS -> Image
32 ;; render ...
33 ;; !!!
34 (define (render ws) ...)

```

Depending on which other big-bang options you are using you would also end up with wish list entries for those handlers. So, at an early stage a world program might look like this:

```

1  (require 2htdp/universe)
2  (require 2htdp/image)
3
4  ;; A cat that walks across the screen.
5
6  ;; =====
7  ;; Constants:
8
9  (define WIDTH 200)
10 (define HEIGHT 200)
11 (define CAT-IMG (circle 10 "solid" "red")) ; a cat
12 (define MTS (empty-scene WIDTH HEIGHT))

```

```

13
14 ;; =====
15 ;; Data definitions:
16
17 ;; Cat is Number
18 ;; interp. x coordinate of
19 ;; cat (in screen coordinates)
20 (define C1 1)
21 (define C2 30)
22
23 #;
24 (define (fn-for-cat c)
25   (... c))
26
27 ;; =====
28 ;; Functions:
29
30 ;; Cat -> Cat
31 ;; start the world with initial state c,
32 ;; for example: (main 0)
33 (define (main c)
34   (big-bang c           ;Cat
35     (on-tick tock)      ;Cat -> Cat
36     (to-draw render))) ;Cat -> Image
37
38 ;; Cat -> Cat
39 ;; Produce cat at next position
40 ;!!!
41 (define (tock c) 1) ;stub
42
43 ;; Cat -> Image
44 ;; produce image with CAT-IMG placed on
45 ;; MTS at proper x, y position
46 ; !!!
47 (define (render c) MTS)

```

Note that we are maintaining a wish list of functions that need to be designed. The way to maintain the wish list is to just write a signature, purpose and stub for each wished-for function, also label the wish list entry with !!! or some other marker that is easy to search for. That will help you find your unfilled wishes later.

Forming wish list entries this way is enough for main (or other functions that call a wished for function) to be defined without error. But of course main (and other such functions) will not run properly until the wished for

functions are actually completely designed.

As you design the program remember to run early and run often. The sooner you can run the program after writing anything the sooner you can find any small mistakes that might be in it. Fixing the small mistakes earlier makes it easier to find any harder mistakes later.

4.4 Key and Mouse Handlers

The `on-key` and `on-mouse handler` function templates are handled specially. The `on-key` function is templated according to its second argument, a `KeyEvent`, using the large enumeration rule. The `on-mouse` function is templated according to its `MouseEvent` argument, also using the large enumeration rule. So, for example, for a `key handler function` that has a special behaviour when the space key is pressed but does nothing for any other key event the following would be the template:

```
1 (define (handle-key ws ke)
2     (cond [(key=? ke " ") (... ws)]
3           [else
4             (... ws)]))
```

Similarly the template for a mouse handler function that has special behavior for `mouse clicks` but ignores all other mouse events would be:

```
1 (define (handle-mouse ws x y me)
2     (cond [(mouse=? me "button-down") (... ws x y)]
3           [else
4             (... ws x y)]))
```

For more information on the `KeyEvent` and `MouseEvent` large enumerations see the DrRacket help desk.

Chapter 5

Data Driven Templates

Templates are the core structure that we know a function must have, independent of the details of its definition. In many cases the template for a function is determined by the type of data the function consumes. We refer to these as data driven templates. The recipe below can be used to produce a data driven template for any type comment.

For a given type TypeName the data driven template is:

```
1 (define (fn-for-type-name x)
2   <body>)
```

Where x is an appropriately chosen parameter name (often the initials of the type name) and the body is determined according to the table below. To use the table, start with the type of the parameter, i.e. TypeName, and select the row of the table that matches that type. The first row matches only primitive types, the later rows match parts of type comments.

(Note that when designing functions that consume additional atomic parameters, the name of that parameter gets added after every ... in the template. Templates for functions with additional complex parameters are covered in Functions on 2 One-Of Data.)

Type of data	Cond question (if applicable)	Body or cond answer (if applicable)
<i>Atomic Non-Distinct</i> <ul style="list-style-type: none"> • Number • String • Boolean • Image • interval like Number[0, 10) • etc. 	Appropriate predicate <ul style="list-style-type: none"> • (number? x) • (string? x) • (boolean? x) • (image? x) • (and (<= 0 x) (< x 10)) • etc. 	Expression that operates on the parameter. (... x)
<i>Atomic Distinct Value</i> <ul style="list-style-type: none"> • "red" • false • empty • etc. 	Appropriate predicate <ul style="list-style-type: none"> • (string=? x "red") • (false? x) • (empty? x) • etc. 	Since value is distinct, parameter does not appear. (...)

Type of data	Cond question (if applicable)	Body or cond answer (if applicable)
<p><i>One Of</i></p> <ul style="list-style-type: none"> • enumerations • itemizations 		<p>Cond with one clause per subclass of <i>one of</i>.</p> <pre> ; Q - question ; A - answer (cond [<Q1> <A1>] [<Q2> <A2>]) </pre> <p>Where each question and answer expression is formed by following the rule in the question or answer column of this table for the corresponding case. A detailed derivation of a template for a one-of type appears below.</p> <p>It is permissible to use <i>else</i> for the last question for itemizations and large enumerations. Normal enumerations should not use <i>else</i>.</p> <p>Note that in a mixed data itemization, such as:</p> <pre> ;; Measurement is ↪ one of: ;;- Number[-10, 0) ;;- true ;;- Number(0, 10] </pre> <p>the cond questions must be <i>guarded</i></p>

Type of data	Cond question (if applicable)	Body or cond answer (if applicable)
		<p>with an appropriate type predicate. In particular, the first cond question for Measurement must be</p> <pre>(and (number? m) (<= -10 m) (< m 0))</pre> <p>where the call to <code>number?</code> guards the calls to <code><=</code> and <code><</code>. This will protect <code><=</code> and <code><</code> from ever receiving true as an argument.</p>
<i>Compound</i> <ul style="list-style-type: none"> • Position • Firework • Ball • cons • etc. 	Predicate from structure <ul style="list-style-type: none"> • (posn? x) • (firework? x) • (ball? x) • (cons? x) (often just <code>else</code>) • etc. 	All selectors. <ul style="list-style-type: none"> • (... (posn-x ↪ x) (posn-y ↪ x)) • (... ↪ (firework- ↪ y x) ↪ (firework- ↪ color ↪ x)) • (... (ball-x ↪ x) ↪ (ball-dx ↪ x)) • (... (first x) ↪ (rest x)) • etc. <p>Then consider the</p>

Type of data	Cond question (if applicable)	Body or cond answer (if applicable)
		<p>result type of each selector call and wrap the accessor expression appropriately using the table with that type. So for example, if after adding all the selectors you have:</p> <pre>(... (game-ball g) ↪ ;produces Ball (game-paddle g)) ↪ ;produces Paddle</pre> <p>Then, because both Ball and Paddle are non-primitive types (types that you yourself defined in a data definition) the reference rule (immediately below) says that you should add calls to those types' template functions as follows:</p> <pre>(... (fn-for-ball ↪ (game-ball g)) (fn-for-paddle ↪ (game-paddle ↪ g)))</pre>
Other Non-Primitive Type Reference	Predicate, usually from structure definition <ul style="list-style-type: none"> • (firework? x) • (person? x) 	Call to other type's template function <ul style="list-style-type: none"> • (fn-for-firework ↪ firework ↪ x) • (fn-for-person ↪ x)

Type of data	Cond question (if applicable)	Body or cond answer (if applicable)
<i>Self Reference</i>		Form natural recursion with call to this type's template function: <pre>(fn-for-los (rest ↪ los))</pre>
<i>Mutual Reference</i> Note: form and group all templates in mutual reference cycle together.		Call to other type's template function: <pre>(fn-for-lod ↪ (dir-subdirs ↪ d) (fn-for-dir (first ↪ lod))</pre>

Table 5.1: Data Driven Templates

5.1 Producing the Template for an Example - One Of Type

In many cases more than one of the above rules will apply to a single template. Consider this type comment:

```
1  ;; Clock is one of:
2  ;; - Natural
3  ;; - false
```

and the step-by-step construction of the template for a function operating on Clock.

1. Clock is a one of type with two subclasses (one of which is not distinct making it an itemization). The one of rule tells us to use a cond. The cond needs one clause for each subclass of the itemization.

```
1  (define (fn-for-clock c)
2      (cond [Q A]
3            [Q A]))
4
5  ;; Template rules used:
6  ;; - one of: 2 cases
```

2. The cond questions need to identify each subclass of data. The cond answers need to follow templating rules for that subclasses data. In the first subclass, Natural is a non-distinct type; the atomic non-distinct rule tells us the question and answer as shown to the left.

```
1  (define (fn-for-clock c)
2      (cond [(number? c) (... c)]
3            [Q A]))
4
5  ;; Template rules used:
6  ;; - one of: 2 cases
7  ;; - atomic non-distinct: Natural
```

3. In the second case false is an atomic distinct type, so the atomic-distinct rule gives us the question and answer. Since the second case is also the last case we can use else for the question.

```
1  (define (fn-for-clock c)
2      (cond [(number? c) (... c)]
3            [else
4              (...)]))
5  ;; Template rules used:
```

```

6  ;; - one of: 2 cases
7  ;; - atomic non-distinct: Natural
8  ;; - atomic distinct: false

```

5.2 Templates for Mutually Referential Types

The previous example doesn't cover the mutual-reference rule, which says that in the case of mutually-referential data definitions, when you template one function in the mutual-reference cycle you should immediately template all the functions in the mutual-reference cycle. So, for example, given:

```

1  (define-struct person (name subs))
2  ;; Person is (make-person String ListOfPerson)
3
4  ;; ListOfPerson is one of:
5  ;; - empty
6  ;; - (cons Person ListOfPerson)

```

Then if you need a template for a function operating on a Person (or a function operating on a ListOfPerson) you should immediately write a template for both functions, resulting in something like this:

```

1  #;
2  (define (fn-for-person p)
3      (... (person-name p)
4            (fn-for-lop (person-subs p)))) ;MRMR
5
6  #;
7  (define (fn-for-lop lop)
8      (cond [(empty? lop) ...]
9            [else
10             (... (fn-for-person (first lop)) ;MRMR
11                  (fn-for-lop (rest lop)))])) ;NRSF
12
13 ; - MRMR: mutual recursion from mutual-reference
14 ; - NRSF: natural recursion from self-reference

```

(Note that producing that template will also involve using the atomic-distinct, atomic, one-of and compound rules.)

As with self-reference, its a good idea to draw a mutual-reference line on the data definition and ensure you have corresponding mutual recursion lines in your templates.

5.3 Testing

The principles above can also be used to understand how many tests a data definition implies. Simply put, the set of tests/examples should cover all cases, call all helper functions, involve all selectors, and avoid duplicated values.

5.4 Additional Design Rules for Helpers

During coding three additional guidelines suggest situations under which a helper function should be added:

1. Use a separate function for each difference between quantities in a problem.
2. If a subtask requires operating on arbitrary sized data a helper function must be used.
3. If a subtask involves special domain knowledge a helper function should be used.
4. In addition always keep the "one task per function" goal in mind. If part of a function you are designing seems to be a well-defined subtask put that into a helper function.

Chapter 6

Functions On 2 One-Of Data

This chapter outlines the variations in the normal HtDF recipe when designing a function that consumes 2 data that have a one-of in their type comments. Examples of functions for which this applies include functions with the following signatures:

```
1  ;; ListOfString ListOfString -> Boolean
2  ;; ListOfString ListOfString -> ListOfString
3  ;; ListOfString BinaryTree -> Boolean
4  ;; ListOfNatural FamilyTree -> ListOfString
```

For the purpose of this explanation, assume that the goal is to design a function that consumes two ListOfString and produces true if the strings in the first list are equal to the corresponding strings in the second list. If that is true and the second list is longer than the first the function should produce true; if the second list is shorter it always produces false.

The first three steps of the recipe - signature, purpose and stub - are unchanged.

```
1  ;; ListOfString ListOfString -> Boolean
2  ;; produce true if lsta is a prefix of lstb
3
4  (define (prefix=? lsta lstb) false)
```

But at this point the next step is to form a *cross-product of type comments* table as follows. The row labels of the table are the cases of the one-of type comment for one argument (perhaps the first), and the column labels are the cases of the one-of type comment for the other argument.

```

1  ;; CROSS PRODUCT OF TYPE COMMENTS TABLE
2  ;;
3  ;;               empty      lstb      (cons String LOS)
4  ;;               /
5  ;; l  empty      /
6  ;; s  -----
7  ;; t  (cons String LOS) /
8  ;; a  /

```

In this case, where both arguments have 2 cases in their one-of type comments, the cross-product formed has 4 cells ($2 * 2 = 4$). The next step of the process is to use the cross product table to help form at least as many tests as there are cells. The upper left cell describes a scenario where both `lsta` and `lstb` are empty. The lower left cell is where `lsta` is non-empty but `lstb` is empty and so on. The lower right cell is where both lists are non-empty, and this case requires more than one test. So we end up with:

```

1  (check-expect (prefix=? empty empty) true)
2  (check-expect (prefix=? empty (list "a" "b")) true)
3  (check-expect (prefix=? (list "a") empty) false)
4
5  (check-expect (prefix=? (list "a") (list "b")) false)
6  (check-expect (prefix=? (list "a") (list "a")) true)
7  (check-expect (prefix=? (list "a" "b") (list "a" "x"))
8  false)
9  (check-expect (prefix=? (list "a" "b") (list "a" "b"))
10 true)

```

We can now use the tests to help fill in the contents of the table cells, indicating what the function should do in each case. Any cell requiring a more complex code answer, such as the lower right one in this case, need not be coded perfectly with correct syntax, but should give a good idea of what the code must do.

```

1  ;; CROSS PRODUCT OF TYPE COMMENTS TABLE
2  ;;
3  ;;               empty      lstb      (cons String LOS)
4  ;;               /
5  ;; l  empty      true  / true
6  ;; s  -----
7  ;; t  (cons String LOS) false / (and
8  ;; a  /      <firsts are string=?>
9  ;;      /      <rests are prefix=?>)

```

Now comes the most fun step. We look for a way to simplify the table by identifying cells that have the same answer. In this case the entire first

row produces true, so we can simplify the table by combining the two cells in the first row into a single cell:

```

1  ;; CROSS PRODUCT OF TYPE COMMENTS TABLE
2  ;;                                lstb
3  ;;                                empty    (cons String LOS)
4  ;;                                /
5  ;; l  empty                        true
6  ;; s                                -----
7  ;; t  (cons String LOS)  false  / (and
8  ;; a                                /  <firsts are string=?>
9  ;;                                /  <rests are prefix=?>)

```

Now we are almost done. The next step is to code the function body directly from the table. This in effect intertwines templating with coding of details. Because the simplified table has only three cells, we know that the body of the function will be a three case cond. For the first question we always pick the largest cell, in this case the top row. The question needs to be true of the entire combined cell, so in this case the question is (empty? lsta). The answer in this case is just true.

For the next case of the cond we pick the lower left cell. At this point, we know that the top row has been handled, so we only need a question that distinguishes the remaining cells apart. In this case (empty? lstb) distinguishes the lower left from the lower right cell. The answer in this case is false.

In larger tables this process continues until you get to the last cell, at which point the question can be else.

In cells that involve natural recursion, the natural recursion can be formed by applying the normal rules for handling self-reference. In this case the template for the cond answer in the third cond case is:

```

1  (... (first lsta)
2       (first lstb)
3       (prefix=? (rest lsta) ...))
4       (prefix=? ... (rest lstb)))

```

After filling that in we end up with:

```

1  (define (prefix=? lsta lstb)
2    (cond [(empty? lsta) true]
3          [(empty? lstb) false]
4          [else
5           (and (string=? (first lsta) (first lstb))
6                (prefix=? (rest lsta)
7                           (rest lstb)))]))

```

In general, when designing a function on 2 one of data it is a good idea to include the cross-product table in the design.

Chapter 7


Control Driven Templates

7.1 Function Composition

Use function composition when a function must perform two or more distinct and complete operations on the consumed data. For example:

- A function that must sort and layout a list of images. First it must sort the complete list and then lay it out. It cannot sort and layout each image one at a time.
- A function that must advance a list of raindrops and then remove the ones that have left the screen. First it must advance all the drops and then remove the ones that have advanced too far. (With difficulty this could be done in a single pass through the list of drops, but it is much more cumbersome to do that way.)

When using function composition the normal template for the function is discarded, and the body of the function has two or more function compositions. So in the case of arrange-images the function design would look like this:



```
1  ;; ListOfImage -> Image
2  ;; arrange images left to right in increasing order of size
3  (check-expect (arrange-images (list I1 I3 I2))
4                (beside I1 I2 I3 BLANK))
5
6  (define (arrange-images loi)
7        (layout-images (sort-images loi)))
```

Which we read as saying "first sort the images, and then layout the sorted list". At the point this is written wish list entries would be created for layout-images and sort-images unless those functions already existed.

Tests for a function that uses function composition should be selected to ensure that the function is calling all the appropriate functions and composing them properly. For example, assuming that images I1, I2 and I3 are in increasing order of size, then this test alone would not be adequate:

```
1 (check-expect (arrange-images (list I1 I2 I3))
2               (beside I1 I2 I3 BLANK))
```

Because a faulty implementation of `arrange-images` that just calls `layout-images` would pass. Instead a test like the original one above is needed, to ensure that both `sort-images` and `layout-images` are called. But note that the tests for `arrange-images` do not themselves need to fully test both composed functions. They only need to test the composition. That is why `arrange-images` does not absolutely have to have a base case test. (Although it wouldn't hurt it to have one.)

7.2 Backtracking Search

The template for backtracking is:

```
1 (define (fn-for-x x)
2   (... (fn-for-lox (x-subst x))))
3
4 (define (fn-for-lox lox)
5   (cond [(empty? lox) false]
6         [else
7          ;is first child successful?
8          (if (not (false? (fn-for-x (first lox))))
9              ;if so produce that
10             (fn-for-x (first lox))
11             ;or try rest of children
12             (fn-for-lox (rest lox)))]))
```

Note that this template incorporates the template for an n-ary tree, where the tree nodes have type `x` and `x-subst` produces the children of a tree node given the node. The backtracking works as commented above. Once we have local expressions we tend to write the backtracking search template as follows:

```
1 (define (backtracking-fn x)
2   (local [(define (fn-for-x x)
3             (... (fn-for-lox (x-subst x))))
4
5             (define (fn-for-lox lox)
```

```

6      (cond [(empty? lox) false]
7            [else
8              (local
9                [(define try
10                  (fn-for-x
11                    (first lox)))]
12                (if (not (false? try))
13                    try
14                    (fn-for-lox
15                      (rest lox))))))]
16      (fn-for-x x)))

```

7.3 Generative Recursion

The template for generative recursion is:

```
1  (define (genrec-fn d)
2      (cond [(trivial? d) (trivial-answer d)]
3            [else
4              (... d
5                (genrec-fn (next-problem d)))]))
```


Chapter 8

Accumulators

There are three general ways to use an accumulator in a recursive function (or set of mutually recursive functions):

1. To preserve context otherwise lost in structural recursion.
2. To make a function tail recursive by preserving a representation of the work done so far (aka a result-so-far accumulator).
3. To make a function tail recursive by preserving a representation of the work remaining to do (aka a worklist accumulator).

The same basic recipe covers all three forms of accumulators. The main example shown here is a context preserving accumulator.

8.1 The basic recipe and context preserving accumulators.

Signature, purpose, stub and examples

Design of the function begins normally, with signature, purpose, stub and examples.

```
1  ;; (listof X) -> (listof X)
2  ;; produce list formed by keeping
3  ;; the 1st, 3rd, 5th and so on elements of lox
4  (check-expect (skip1 (list "a" "b" "c" "d"))
5                (list "a" "c"))
6  (check-expect (skip1 (list 0 1 2 3 4)) (list 0 2 4))
7
8  (define (skip1 lox) empty) ;stub
```

Templating

The template step is a 3 part process. The first step is to template normally according to the rules for structural recursion, i.e. template according to the (listof X) parameter.

```
1 (define (skip1 lox)
2     (cond [(empty? lox) (...)]
3           [else
4             (... (first lox)
5                 (skip1 (rest lox)))]))
```

The next step is to encapsulate that function in an outer function and local. As part of this step give the outer function parameter a different name than the inner function parameter. Note that if you are working with multiple mutually recursive functions they call get wrapped in a single outer function.

```
1 (define (skip1 lox0)
2     (local [(define (skip1 lox)
3                 (cond [(empty? lox) (...)]
4                       [else
5                         (... (first lox)
6                             (skip1 (rest lox)))]))]
7       (skip1 lox0)))
```

Now add the accumulator parameter to the inner function. In addition, add ... or more substantial template expressions in each place that calls the inner function. During this step treat the accumulator parameter as atomic.

```
1 (define (skip1 lox0)
2     (local [(define (skip1 lox acc)
3                 (cond [(empty? lox) (... acc)]
4                       [else
5                         (... acc
6                             (first lox)
7                             (skip1 (rest lox)
8                                   (... acc
9                                       (first lox)))]))]
10       (skip1 lox0 ...)))
```

Accumulator type, invariant and examples

The next step is to work out what information the accumulator will represent and how it will do that. Will the accumulator serve to represent some

context that would otherwise be lost to structural recursion? Or, to support tail recursion will it represent some form of result so far? Or will it represent a work list of some sort. In many cases this is clear before reaching this stage of the design process. In other cases examples can be used to work this out. But in all cases examples are useful to work out exactly how the accumulator will represent the information.

In this case we need to know, in each recursive call to the inner function, whether the current first item in the list should be kept or skipped. There are many ways to represent this, but one simple way is to use a natural that represents how far into the original list (lox0) we have traveled. If we assume the call to the top-level definition of skip1 is (skip1 (list 0 1 2 3 4)), then the progression of calls to the internal skip1 would be as follows:

```

1  ;the 0 is the 1st element of lox0
2  ;(using 1 based indexing)
3  (skip1 (list 0 1 2 3 4) 1)
4  (skip1 (list 1 2 3 4) 2) ;the 1 is the 2nd element
5  (skip1 (list 2 3 4) 3) ;the 2 is the 3rd
6  (skip1 (list 3 4) 4) ;the 3 is the 4th
7  (skip1 (list 4) 5) ;the 4 is the 5th
8  (skip1 (list ) 6)

```

Note that the accumulator value is not constant, but it always represents the position of the current (first lox) in the original list lox0. (You can see here why we renamed the parameter to the outer function, it makes it easier to describe the relation between the original value lox0 and the value in each recursive call lox.)

These examples of the progression of calls to the internal recursive function(s) allow us to work out clearly the accumulator type, as well as its invariant, which describes what is constant about the accumulator as it changes in other words, what property it always represents. In this case the type is Natural, and the invariant is the 1 based index of (first lox) in lox0. So when the accumulator is an odd number we will keep (first lox) in the result, when it is even we will skip (first lox).

```

1  (define (skip2 lox0)
2      ;; acc is Natural
3      ;how many elements of lox to keep before next skip
4      ;; (skip2 (list 0 1 2 3 4) 1)
5      ;; (skip2 (list 1 2 3 4) 0)
6      ;; (skip2 (list 2 3 4) 1)
7      ;; (skip2 (list 3 4) 0)
8      ;; (skip2 (list 4) 1)
9      ;; (skip2 (list ) )

```

```

10
11      (local [(define (skip2 lox acc)
12                (cond [(empty? lox) (... acc)]
13                      [else
14                        (... acc
15                          (first lox)
16                          (skip2 (rest lox)
17                                (... acc
18                                  (first lox))))))]
19      (skip2 lox0 ...)))

```

Complete the code

At this point the function definition can be completed by using the signature, purpose, examples, accumulator type, invariant and examples to fill in the details. When doing so, note three distinct aspects of coding with the accumulator invariant. Initializing the accumulator happens in the trampoline, and involves providing an initial value of the accumulator that satisfies the invariant. Exploiting the invariant involves counting on the accumulator to always represent what the accumulator describes. Preserving the invariant happens in recursive calls to the function where a (possibly) new value is provided for the accumulator argument. Preserving the invariant means making sure that the value provided in the recursive call satisfies the invariant.

```

1  (define (skip1 lox0)
2    ;; acc is Natural
3    ;; 1 based index of (first lox) in lox0
4    ;; (skip1 (list 0 1 2 3 4) 1) ;0 should be kept
5    ;; (skip1 (list 1 2 3 4) 2) ;1 should be skipped
6    ;; (skip1 (list 2 3 4) 3) ;0 should be kept
7    ;; (skip1 (list 3 4) 4) ;1 should be skipped
8    ;; (skip1 (list 4) 5) ;0 should be kept
9    ;; (skip1 (list ) 6)
10   (local [(define (skip1 lox acc)
11             (cond [(empty? lox) empty]
12                   [else
13                     (if (even? acc)
14                         (skip1 (rest lox) (add1 acc))
15                         (cons (first lox)
16                              (skip1 (rest lox)
17                                    (add1 acc))))))]
17   (skip1 lox0 1)))
18

```

8.2 Tail Recursion

To make the function tail recursive, all recursive calls must be in tail position. For functions that operate on flat structures (data that has only one reference cycle), this can be accomplished by using an accumulator to represent build up information about the final result through the series of recursive calls. We often name these accumulators `rsf` but it is worth noting that some functions require more than one result so far accumulator (see `average`).

Making functions that operate on data with more than one cycle in the graph (such as arbitrary-arity trees) usually requires the use of an accumulator to build up the data that still remains to be operated on. This is a worklist accumulator, often called `todo`.

Chapter 9

Template Blending

To understand template blending its important to understand that templates are what we know about the core of a function (or set of functions) before we get to the details. Data driven (or structural recursion) templates are that, backtracking templates are that, and generative recursion templates are that.

In some cases we know more than one thing about the core structure of a function (or set of functions). In the sudoku-solver for example three different templates apply to the solve functions.

arbitrary-arity tree - we consider each board to have a set of next boards formed by filling the first empty cell with the numbers from 1 - 9. So this forms an arbitrary-arity tree (the arity is actually [0, 9]).

generative recursion - while each board has a set of next boards that set is included in the representation of the board. Instead those next boards have to be generated. So we have a generated arbitrary-arity tree.

backtracking search - in addition we need to do a backtracking search over the arbitrary-arity tree.

In template blending we take multiple templates that contribute to the structure of a function (or functions) and combine them together.

We have the following for the solve function:

```
1  ;; Board -> Board or false
2  ;; produce a solution for bd; false if bd is unsolvable
3  ;; Assume: bd is valid
4  (check-expect (solve BD4) BD4s)
5  (check-expect (solve BD5) BD5s)
6  (check-expect (solve BD7) false)
7
8  (define (solve bd) false) ;stub
```

Now let's start with the template for arbitrary-arity tree. Remember that the template for an arbitrary-arity tree involves a mutual recursion. In this case, we would have a function that consumes Board (ie. `solve-bd`) and calls another function (ie. `solve-lobd`) that does something to the (listof Board) that is supposed to come with the Board (a Board doesn't actually come with a (listof Board) but we will deal with it using generative recursion later). In order to complete the mutual recursion, we need to call `solve-bd` inside the function `solve-lobd`. So now we have the following template:

```

1 (define (solve bd)
2     (local [(define (solve--bd bd)
3                 (... (solve--lobd (bd-subs bd))))
4
5                 (define (solve--lobd lobd)
6                     (cond [(empty? lobd) (...)]
7                           [else
8                             (...
9                               (solve--bd (first lobd))
10                                (solve--lobd
11                                  (rest lobd))))))]
12     (solve--bd bd)))

```

Keep in mind that the `bd-subs` selector doesn't exist because Board does not keep a (listof Board). We need to deal with it using generative recursion.

Let's blend this template with the generative recursion template. The generative recursion template looks like this:

```

1 (define (genrec-fn d)
2     (if (trivial? d)
3         (trivial-answer d)
4         (... d
5             (genrec-fn (next-problem d)))))

```

Since we know that `solve-bd` must generate a (listof Board) to pass to `solve-lobd`, we must blend the template into the `solve-bd` function. Then we would have:

```

1 (define (solve bd)
2     (local [(define (solve--bd bd)
3                 (if (solved? bd)
4                     bd
5                     (solve--lobd (next-boards bd))))
6
7                 (define (solve--lobd lobd)
8                     (cond [(empty? lobd) (...)]

```

```

9                                     [else
10                                     (...
11                                     (solve--bd (first lobd))
12                                     (solve--lobd
13                                     (rest lobd)))))]))
14      (solve--bd bd)))

```

Notice the following:

- bd Subs is changed to next-boards because it suggests that we are generating new boards.
- trivial? from the generative recursion template is changed to solved? because the trivial case (ie. the case where the recursion should stop at) is when the board is solved.
- trivial-answer from the generative recursion template is omitted because in this case, the trivial answer is the board that is solved, which is represented by bd.

Lastly, we need to blend in the template for backtracking search. The template is:

```

1  (define (backtracking-fn x)
2      (local [(define (fn-for-x x)
3                  (... (fn-for-lox (x Subs x))))
4                  (define (fn-for-lox lox)
5                      (cond [(empty? lox) false]
6                            [else
7                             (local [(define try
8                                     (fn-for-x (first lox)))]
9                                 (if (not (false? try))
10                                     try
11                                     (fn-for-lox
12                                     (rest lox))))))]
12                      (fn-for-x x)))
13      (fn-for-x x))

```

We need to blend the template for fn-for-lox from above into the function solve-lobd.

```

1  (define (solve bd)
2      (local [(define (solve--bd bd)
3                  (if (solved? bd)
4                      bd
5                      (solve--lobd (next-boards bd))))
6      (solve--bd bd))

```



```

7      (define (solve--lobd lobd)
8        (cond [(empty? lobd) false]
9              [else
10               (local
11                  [(define try
12                     (solve--bd (first lobd)))]
13                    (if (not (false? try))
14                        try
15                        (solve--lobd
16                          (rest lobd))))))]
17      (solve--bd bd)))

```

The template allows the solve-lobd function to do the following:

- If we have reached the end of the (listof Board) (ie. if lobd is empty - the base case), then produce false, meaning that the (listof Board) has no solution.
- If the (listof Board) is not empty, then try to solve the first board in the list (ie. (solve-bd (first lobd))).
- If the outcome of the try is not false (ie. (not (false? try))), then produce that outcome because it means the board has been solved.
- If the outcome is false, then recurse and try the rest of the list (ie. (solve-lobd (rest lobd))).

Now we have the complete template below.

```

1 (define (solve bd)
2   (local [(define (solve--bd bd)
3             (if (solved? bd)
4                 bd
5                 (solve--lobd (next-boards bd))))
6
7         (define (solve--lobd lobd)
8           (cond [(empty? lobd) false]
9                 [else
10                  (local [(define try
11                           (solve--bd
12                             (first lobd)))]
13                    (if (not (false? try))
14                        try
15                        (solve--lobd
16                          (rest lobd))))))]
17   (solve--bd bd)))

```

The next step to completing the function is to implement the two helper functions - `solved?` and `next-boards`.

Chapter 10

Abstraction

10.1 Abstraction From Examples

We design abstract functions in the opposite order of the normal HtDF recipe. We always want to do the easiest thing first, and with the abstract function design processes getting the working function definition is the easiest thing to do. In fact going through the recipe in the opposite order exactly goes from easiest to hardest.

1. Identify two or more fragments of highly repetitive code. In general these can be expressions that appear within functions, or they can be entire functions. The rest of this recipe is tailored to the case where entire functions have been chosen.
2. Arrange the two functions so that it is easy to see them at the same time.
3. Identify one or more points where the functions differ (points of variance). Do not count differences in function names or parameter names as points of variance.
4. Copy one function definition to make new one
 - give the new function a more general name
 - add a new parameter for each point of variance
 - update any recursive calls to use new name and add new parameters in recursive calls
 - use the appropriate new parameter at each point of variability
 - rename other parameters to be more abstract (lon to lox for example)
5. Adapt tests from original functions to new abstract function

- be sure to test variability
 - attempt to test behavior of the abstract function beyond that exercised by the two examples
6. Develop an appropriately abstract purpose based on the examples.
 7. Develop an appropriate signature for the abstract function; in many cases the signature will include type parameters.
 8. Rewrite the body of the two original functions to call the abstract function.

10.2 Abstraction From Type Comments

We design abstract functions in the opposite order of the normal HtDF recipe. We always want to do the easiest thing first, and with the abstract function design processes getting the working function definition is the easiest thing to do. In fact going through the recipe in the opposite order exactly goes from easiest to hardest.

1. if there are templates for mutually recursive functions first encapsulate them in a single template with local
2. replace each ... in the templates with a new parameter; for (...) remove the parens
3. develop examples (check-expects)
4. develop abstract purpose from examples
5. develop abstract signature from concrete examples

Let's now go through generating a fold function for (listof X). Here is the type comment for (listof X):

```

1  ;; ListOfX is one of:
2  ;; - empty
3  ;; - (cons X ListOfX)

```

We can generate the following template based on this type comment:

```

1  (define (fn-for-lox lox)
2      (cond [(empty? lox) (...)]
3            [else
4             (... (first lox)
5                  (fn-for-lox (rest lox)))]))

```

Since there is no mutual recursion in the template, we can skip step 1. According to step 2, we need to replace each ... with a new parameter. We also need to rename the function to fold. Here is what we would have after step 2:

```

1  (define (fold fn b lox)
2      (cond [(empty? lox) b]
3            [else
4              (fn (first lox)
5                  (fold fn b (rest lox)))]))

```

Note that we called the ... in place of the base case is named to b, and the ... in place of the function is named to fn.

For step 3, we should write some examples to test for the function and also find out what we can do with the fold function:

```

1  ; sum of the numbers in the list
2  (check-expect (fold + 0 (list 1 2 3)) 6)
3
4  ; product of the numbers in the list
5  (check-expect (fold * 1 (list 1 2 3)) 6)
6
7  ; append the strings in the list
8  (check-expect (fold string-append ""
9                    (list "a" "bc" "def")) "abcdef")
10
11 ; sum of the areas of the images in the list
12 (check-expect (local [(define (total-area i a)
13                         (+ (* (image-width i)
14                               (image-height i))
15                             a))]
16                 (fold total-area 0 (list
17                                   (rectangle 20 40 "solid" "red")
18                                   (right-triangle 10 20 "solid"
19                                                    ↪ "red")
19                                   (circle 20 "solid" "red"))))
20               (+ (* 20 40) (* 10 20) (* 40 40)))

```

Step 4 requires us to develop an abstract purpose from the examples. In this example, we can just write:

```

1  ;; the abstract fold function for (listof X)

```

In the last step, we need to determine the abstract signature from the examples that we have written. The more diverse the examples that we have, the easier it is to come up with the correct signature.

From the function definition for fold, we can first reason the following for the signature of fold:

- the parameter lox is of the type (listof X)
- the parameter b is of the same type as what the fold function produces because it is the base case
- the predicate function fn takes in two arguments
- the predicate function fn produces the same type as what the fold function produces
- the first parameter for the predicate function fn is of the type X
- the second parameter for the predicate function fn is of the same type as what the fold function produces

From the above reasonings, we can narrow the signature down to:

```
1  ;; (X ??? -> ???) ??? (listof X) -> ???
```

Note that all of the ??? are of the same type according to the above reasoning. Now the question is whether ??? has to be the same type as X or can be something else (ie. Y). From the fourth example in step 3, we can see that the signature for the predicate function total-area is:

```
1  ;; Image Number -> Number
```

In this case, the X is Image and the ??? is Number. This is an example of ??? being something different than X. Therefore, we can conclude that the abstract signature for fold is:

```
1  ;; (X Y -> Y) Y (listof X) -> Y
```

10.3 Using Abstract Functions

The template for using a built-in abstract function like filter is:

```
1  ;; (listof Number) -> (listof Number)
2
3  ;; produce only positive? elements of lon
4
5  ;; tests elided
6
7  ;; template as call to abstract function
8  (define (only-positive lon)
9    (filter ... lon))
```

Now we note that the type of `lon` is `(listof Number)`; and the signature of `filter` is `(X -> Boolean) (listof X) -> (listof X)`. This means that that signature of the function passed to `filter` is `(Number -> Boolean)` so we can further decorate the template as follows:

```
1 (define (only-positive lon)
2       ;(Number -> Boolean)
3       (filter ... lon))
```

as a note to ourselves about the signature of the function we replace `...` with.